NAG Fortran Library Routine Document

F07HSF (CPBTRS/ZPBTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07HSF (CPBTRS/ZPBTRS) solves a complex Hermitian positive-definite band system of linear equations with multiple right-hand sides, $AX = B$, where A has been factorized by F07HRF (CPBTRF/ZPBTRF).

2 Specification

```
SUBROUTINE F07HSF(UPLO, N, KD, NRHS, AB, LDAB, B, LDB, INFO)
ENTRY      cpbtrs (UPLO, N, KD, NRHS, AB, LDAB, B, LDB, INFO)
INTEGER            N, KD, NRHS, LDAB, LDB, INFO
complex            AB(LDAB,*), B(LDB,*)
CHARACTER*1        UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

To solve a complex Hermitian positive-definite band system of linear equations $AX = B$, this routine must be preceded by a call to F07HRF (CPBTRF/ZPBTRF) which computes the Cholesky factorization of A. The solution X is computed by forward and backward substitution.

If UPLO = 'U', $A = U^H U$, where U is upper triangular; the solution X is computed by solving $U^H Y = B$ and then $UX = Y$.

If UPLO = 'L', $A = LL^H$, where L is lower triangular; the solution X is computed by solving $LY = B$ and then $L^H X = Y$.

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: UPLO – CHARACTER*1 *Input*

On entry: indicates whether A has been factorized as $U^H U$ or LL^H as follows:

if UPLO = 'U', $A = U^H U$, where U is upper triangular;

if UPLO = 'L', $A = LL^H$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N $\geq$ 0.

3: KD – INTEGER *Input*

On entry: k, the number of super-diagonals or sub-diagonals of the matrix A.

Constraint: KD ≥ 0.

4: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides.

Constraint: NRHS ≥ 0.

5: AB(LDAB,*) – ***complex*** array *Input*

Note: the second dimension of the array AB must be at least $\max(1, \mathrm{N})$.

On entry: the Cholesky factor of A, as returned by F07HRF (CPBTRF/ZPBTRF).

6: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07HSF (CPBTRS/ZPBTRS) is called.

Constraint: LDAB $\geq$ KD $+ 1$.

7: B(LDB,*) – ***complex*** array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$.

On entry: the n by r right-hand side matrix B.

On exit: the n by r solution matrix X.

8: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07HSF (CPBTRS/ZPBTRS) is called.

Constraint: LDB $\geq \max(1, \mathrm{N})$.

9: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

For each right-hand side vector b, the computed solution x is the exact solution of a perturbed system of equations $(A + E)x = b$, where

$|E| \leq c(k+1)\epsilon |U^H|\,|U|$, if UPLO = 'U',

$|E| \leq c(k+1)\epsilon |L|\,|L^H|$, if UPLO = 'L',

$c(k+1)$ is a modest linear function of $k+1$, and ϵ is the ***machine precision***.

If $\hat{x}$ is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_\infty}{\|x\|_\infty} \leq c(k+1)\,\mathrm{cond}(A, x)\epsilon$$

where $\mathrm{cond}(A,x) = \||A^{-1}|\,|A|\,|x|\|_\infty / \|x\|_\infty \leq \mathrm{cond}(A) = \||A^{-1}|\,|A|\|_\infty \leq \kappa_\infty(A)$. Note that $\mathrm{cond}(A,x)$ can be much smaller than $\mathrm{cond}(A)$.

Forward and backward error bounds can be computed by calling F07HVF (CPBRFS/ZPBRFS), and an estimate for $\kappa_\infty(A)$ $(= \kappa_1(A))$ can be obtained by calling F07HUF (CPBCON/ZPBCON).

8 Further Comments

The total number of real floating-point operations is approximately $16nkr$, assuming $n \gg k$.

This routine may be followed by a call to F07HVF (CPBRFS/ZPBRFS) to refine the solution and return an error estimate.

The real analogue of this routine is F07HEF (SPBTRS/DPBTRS).

9 Example

To solve the system of equations $AX = B$, where

$$A = \begin{pmatrix} 9.39 + 0.00i & 1.08 - 1.73i & 0.00 + 0.00i & 0.00 + 0.00i \\ 1.08 + 1.73i & 1.69 + 0.00i & -0.04 + 0.29i & 0.00 + 0.00i \\ 0.00 + 0.00i & -0.04 - 0.29i & 2.65 + 0.00i & -0.33 + 2.24i \\ 0.00 + 0.00i & 0.00 + 0.00i & -0.33 - 2.24i & 2.17 + 0.00i \end{pmatrix}$$

and

$$B = \begin{pmatrix} -12.42 + 68.42i & 54.30 - 56.56i \\ -9.93 + 0.88i & 18.32 + 4.76i \\ -27.30 - 0.01i & -4.40 + 9.97i \\ 5.31 + 23.63i & 9.43 + 1.41i \end{pmatrix}.$$

Here A is Hermitian positive-definite, and is treated as a band matrix, which must first be factorized by F07HRF (CPBTRF/ZPBTRF).

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07HSF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KDMAX, LDAB, NRHMAX, LDB
      PARAMETER        (NMAX=8,KDMAX=8,LDAB=KDMAX+1,NRHMAX=NMAX,
     +                 LDB=NMAX)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, KD, N, NRHS
      CHARACTER        UPLO
*     .. Local Arrays ..
      complex          AB(LDAB,NMAX), B(LDB,NRHMAX)
      CHARACTER        CLABS(1), RLABS(1)
*     .. External Subroutines ..
      EXTERNAL         cpbtrf, cpbtrs, X04DBF
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07HSF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KD, NRHS
      IF (N.LE.NMAX .AND. KD.LE.KDMAX .AND. NRHS.LE.NRHMAX) THEN
*
*        Read A and B from data file
*
```

```
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            DO 20 I = 1, N
               READ (NIN,*) (AB(KD+1+I-J,J),J=I,MIN(N,I+KD))
   20       CONTINUE
         ELSE IF (UPLO.EQ.'L') THEN
            DO 40 I = 1, N
               READ (NIN,*) (AB(1+I-J,J),J=MAX(1,I-KD),I)
   40       CONTINUE
         END IF
         READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*        Factorize A
*
         CALL cpbtrf(UPLO,N,KD,AB,LDAB,INFO)
*
         WRITE (NOUT,*)
         IF (INFO.EQ.0) THEN
*
*           Compute solution
*
            CALL cpbtrs(UPLO,N,KD,NRHS,AB,LDAB,B,LDB,INFO)
*
*           Print solution
*
            IFAIL = 0
            CALL X04DBF('General',' ',N,NRHS,B,LDB,'Bracketed','F7.4',
     +                  'Solution(s)','Integer',RLABS,'Integer',CLABS,
     +                  80,0,IFAIL)
         ELSE
            WRITE (NOUT,*) 'A is not positive-definite'
         END IF
       END IF
       STOP
*
       END
```

9.2 Program Data

```
F07HSF Example Program Data
  4  1  2                                              :Values of N, KD and NRHS
  'L'                                                     :Value of UPLO
 ( 9.39, 0.00)
 ( 1.08, 1.73) ( 1.69, 0.00)
               (-0.04,-0.29) ( 2.65, 0.00)
                             (-0.33,-2.24) ( 2.17, 0.00)  :End of matrix A
 (-12.42,68.42) (54.30,-56.56)
 ( -9.93, 0.88) (18.32,  4.76)
 (-27.30,-0.01) (-4.40,  9.97)
 (  5.31,23.63) ( 9.43,  1.41)                            :End of matrix B
```

9.3 Program Results

```
 F07HSF Example Program Results

 Solution(s)
                      1                   2
 1  (-1.0000, 8.0000) ( 5.0000,-6.0000)
 2  ( 2.0000,-3.0000) ( 2.0000, 3.0000)
 3  (-4.0000,-5.0000) (-8.0000, 4.0000)
 4  ( 7.0000, 6.0000) (-1.0000,-7.0000)
```